# Maintainer: Alexander F. Rødseth <xyproto@archlinux.org>
# Contributor: emersion <contact@emersion.fr>

pkgname=grim
pkgver=1.4.0
pkgrel=3
pkgdesc='Screenshot utility for Wayland'
arch=(loong64 x86_64)
url='https://git.sr.ht/~emersion/grim'
license=(MIT)
depends=(cairo libjpeg-turbo wayland)
makedepends=(fish git meson ninja pango scdoc wayland-protocols)
source=("git+$url#commit=e2cb2d12790fb4d5e91ddc20c3dbe896b00f70ee") # tag: v1.4.0
b2sums=('SKIP')

build() {
  arch-meson build $pkgname
  ninja -C build
}

package() {
  DESTDIR="$pkgdir" ninja -C build install
  install -Dm644 $pkgname/LICENSE \
    "$pkgdir/usr/share/licenses/$pkgname/LICENSE"
}
